

Etude d'un problème ouvert avec Maxima

CALCUL D'UNE AIRE ENTRE DEUX COURBES

1 Enoncé du problème

Calculer une valeur approchée de l'aire comprise entre les courbes représentatives des fonctions f et g définies par :


```
(% i1) f(x):=(x-3)*(x-2)*(x-1)*x*(x+1)*(x+2)*(x+3);  
(%o1) f(x):=(x-3)(x-2)(x-1)x(x+1)(x+2)(x+3)
```

```
(% i2) g(x):=exp(x);  
(%o2) g(x):=exp(x)
```

2 Première approche graphique

On commence par tracer une première esquisse des 2 courbes :


```
(% i3) wxdraw2d(  
  title= ,  
  grid=true,  
  xaxis=true,  
  yaxis=true,  
  xrange=[-10,10],  
  color=black,  
  line_width=1,  
  key= f(x) ,  
  explicit(f(x),x,-10,10),  
  color=blue,  
  key= g(x) ,  
  explicit(g(x),x,-10,10)  
)$
```



```
(%t3)
```

On s'aperçoit donc qu'il faut ajuster les axes et les échelles. On procède par tâtonnements pour réduire l'intervalle où les courbes se croisent à $[-5;7]$ et on ajuste l'axe des ordonnées de -100 à 100 pour que la courbe de f soit visible en entier sur cet intervalle.

```
(% i4) wxdraw2d(
      title= ,
      grid=true,
      xaxis=true,
      yaxis=true,
      xrange=[-5,7],
      yrange=[-100,100],
      color=black,
      line_width=1,
      key= f(x) ,
      explicit(f(x),x,-10,10),
      color=blue,
      key= g(x) ,
      explicit(g(x),x,-10,10)
    )
  $
```



(%t4)

Il semble d'après le graphique que nous devons couper en 6 intervalles sur lesquels les courbes définissent l'aire comprise entre les 2 courbes, sachant que les positions respectives des courbes changent sur chaque intervalle. Pour cela, il nous faut d'abord étudier les points d'intersection entre les deux courbes, c'est à dire trouver les racines et le signe de $h=f-g$.

3 Approche du signe et des racines de h

```
(% i5) h(x) :=f(x)-g(x)$
```

```
(% i6) solve(h(x)=0,x);
```

```
(%o6) [0 = x7 - 14x5 + 49x3 - 36x - %ex]
```

Première constatation : on ne sait pas résoudre l'équation $h(x)=0$. On va donc démontrer qu'il y a 7 racines dont on trouvera des valeurs approchées. Pour cela, il nous faut étudier les variations de h .

```
(% i7) expand(diff(h(x),x));
(%o7) 7x6 - 70x4 + 147x2 - %ex - 36
```

On ne peut trouver les zéros de la dérivée de h (forme identique à h). Une idée est donc de dériver suffisamment pour obtenir une fonction dont on pourra étudier le signe, puis remonter aux zéros et au signe des primitives successives. Définissons ces dérivées successives :

```
(% i8) define(h1(x),expand(diff(h(x),x,1)));
(%o8) h1(x) := 7x6 - 70x4 + 147x2 - %ex - 36
```

```
(% i9) define(h2(x),expand(diff(h(x),x,2)));
(%o9) h2(x) := 42x5 - 280x3 + 294x - %ex
```

```
(% i10) define(h3(x),expand(diff(h(x),x,3)));
(%o10) h3(x) := 210x4 - 840x2 - %ex + 294
```

```
(% i11) define(h4(x),expand(diff(h(x),x,4)));
(%o11) h4(x) := 840x3 - 1680x - %ex
```

```
(% i12) define(h5(x),expand(diff(h(x),x,5)));
(%o12) h5(x) := 2520x2 - %ex - 1680
```

```
(% i13) define(h6(x),expand(diff(h(x),x,6)));
(%o13) h6(x) := 5040x - %ex
```

```
(% i14) define(h7(x),expand(diff(h(x),x,7)));
(%o14) h7(x) := 5040 - %ex
```

On est en capacité d'étudier la dérivée 7ème de h. Nous allons en déduire les comportements des primitives successives de cette dérivée 7ème.

4 Etude de h6, dérivée 6ème de h

```
(% i16) x1 :solve(h7(x)=0,x);float(x1);
(x1) [x = log(5040)]
```

```
(%o16) [x = 8.525161361065415]
```

```
(% i17) x1 :rhs(x1[1]);
(x1) log(5040)
```

Si $x > \log(5040)$ alors $\exp(x) > 5040$ et $h7(x) < 0$. $h7(x)$ est donc positif pour $x < x1$ et négatif pour $x > x1$, ce qui donne le sens de variation de la fonction $h6$:


```
(% i18) 'limit('h6(x),x,minf)=limit(h6(x),x,minf);
(%o18) lim_{x \to -\infty} h6(x) = -\infty
```

```
(% i19) 'limit('h6(x),x,inf)=limit(h6(x),x,inf);
(%o19) lim_{x \to \infty} h6(x) = -\infty
```

```
(% i20) y1 :float(subst(x1,x,h6(x)));
(y1) 37926.81325976969
```

A l'aide de ces résultats, on peut donc tracer le tableau de variation de $h6$. L'objectif étant de trouver les 0 des dérivées, on utilisera donc les valeurs approchées des racines et valeurs des extremas pour simplifier les tableaux de variation.

```
(% i21) wxdraw2d(
title= ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[-5,12],
yrange=[-10000,50000],
color=black,
line_width=1,
key= ,
explicit(h6(x),x,-5,12)
)$
```



(%t21)

x	$-\infty$	8.5	$+\infty$
$h_6(x)$	$-\infty$	37927	$-\infty$

FIGURE 1 – /fig1.png

La fonction h_6 est continue et strictement croissante sur $]-\infty; x_1]$. Les valeurs aux bornes de l'intervalle étant de signe opposé, le théorème des valeurs intermédiaires prouve qu'il existe une unique solution x_2 de l'équation $h_6(x)=0$ sur cet intervalle. De même, il existe un seul x_3 sur l'intervalle $[x_1; +\infty[$ tel que $h_6(x)=0$. D'après les variations de h_6 , $h_6(x) < 0$ sur $]-\infty; x_2[$, > 0 sur $]x_2; x_3[$ et < 0 sur $]x_3; +\infty[$. On cherche une valeur approchée de x_2 et x_3 puis l'on construira le tableau de variations de h_5 à partir de ces éléments. Les raisonnements sur les signes des dérivées successives de h étant similaires, on donnera juste les résultats directement sans expliciter à chaque fois le théorème des valeurs intermédiaires et la lecture des tableaux de variations.

```
(% i23) x2 :find_root(h6(x)=0,x,-10,8);x3 :find_root(h6(x)=0,x,9,20);
(x2) 1.984520777322715210-4
(x3) 10.915329542302088
```

5 Variations de h5

```
(% i27) limit(h5(x),x,minf);h5(x2);h5(x3);limit(h5(x),x,inf);
(%o24) inf
(%o25) - 1681.0000992260382
(%o26) 243550.67502977027
(%o27) - ∞
```




FIGURE 2 – /fig2.png

h_5 s'annule et change de signe en x_4 , x_5 et x_6 :

```
(% i30) x4 :find_root(h5(x)=0,x,-10,0);x5 :find_root(h5(x)=0,x,0,5);x6 :find_root(h5(x)=0,x,10,20);
(x4) - 0.8166039650172688
(x5) 0.8170465112353121
(x6) 12.950263378980932
```

6 Variations de h4

```
(% i35) limit(h4(x),x,minf);h4(x4);h4(x5);h4(x6);limit(h4(x),x,inf);
(%o31) min.f
(%o32) 914.0342169930018
(%o33) - 916.7393520766582
(%o34) 1381675.1710631074
(%o35) - ∞
```




FIGURE 3 – fig3.png

h_4 s'annule et change de signe en x_7 , x_8 , x_9 et x_{10} :

```
(% i39) x7 :find_root(h4(x)=0,x,-10,-1);x8 :find_root(h4(x)=0,x,-0.7,0.8);x9 :find_root(h4(x)=0,x,1,12);x10 :find_root(h4(x)=0,x,10,20);
(x7) - 1.4141411954111884
(x8) - 5.94884208058858310-4
(x9) 1.4154376537098208
(x10) 14.810187797935896
```

7 Variations de h3

```
(% i45) limit(h3(x),x,minf);h3(x7);h3(x8);h3(x9);h3(x10);limit(h3(x),x,inf);
(%o40) inf
(%o41) - 546.2431255310189
(%o42) 293.0002974420602
(%o43) - 550.115768965149
(%o44) 7215456.139048234
(%o45) - ∞
```




FIGURE 4 – /fig4.png

$h3$ s'annule et change de signe en $x11$, $x12$, $x13$, $x14$ et $x15$:

```
(% i50) x11 :find_root(h3(x)=0,x,-3,-1.3);x12 :find_root(h3(x)=0,x,-1.4,0);x13 :find_root(h3(x)=0,x,0,1.3);
x14 :find_root(h3(x)=0,x,1.5,10);x15 :find_root(h3(x)=0,x,15,20);
(x11) - 1.9007030891098804
(x12) - 0.6218969525870028
(x13) 0.6203265166695894
(x14) 1.9032408627054176
(x15) 16.560520025376285
```

8 Variations de h2

```
(% i57) limit(h2(x),x,minf);h2(x11);h2(x12);h2(x13);h2(x14);h2(x15);limit(h2(x),x,inf);
(%o51) minf
(%o52) 321.80948738814845
(%o53) - 119.93537684595222
(%o54) 117.5370348123026
(%o55) - 328.6578578356648
(%o56) 3.548256748926218107
(%o57) - ∞
```




FIGURE 5 – /fig5.png

$h2$ s'annule et change de signe 6 fois en $x16$, $x17$, $x18$, $x19$, $x20$, $x21$:

```
(% i63) x16 :find_root(h2(x)=0,x,-5,-1.9);x17 :find_root(h2(x)=0,x,-1.9,-0.6);x18 :find_root(h2(x)=0,x,-0.7,0.6);
x19 :find_root(h2(x)=0,x,0.7,1.9);x20 :find_root(h2(x)=0,x,1.9,16);x21 :find_root(h2(x)=0,x,17,20);
(x16) - 2.315312170163902
(x17) - 1.1434085725514835
(x18) 0.0034130271777956583
(x19) 1.1356556453175675
(x20) 2.3208887372752707
(x21) 18.23388803600938
```

9 Etude de h1

```
(% i71) limit(h1(x),x,minf);h1(x16);h1(x17);h1(x18);h1(x19);h1(x20);h1(x21);limit(h1(x),x,inf);
(%o64) inf
(%o65) - 181.31617678717225
(%o66) 51.86156219196123
(%o67) - 37.00170650077158
(%o68) 49.05622011738006
(%o69) - 191.37413560144182
(%o70) 1.6661038500080013108
(%o71) - ∞
```




FIGURE 6 – /fig6.png

h1 s'annule et change de signe 7 fois en x22, x23, x24, x25, x26, x27 et x28.

```
(% i78) x22 :find_root(h1(x)=0,x,-5,-2);x23 :find_root(h1(x)=0,x,-2.3,-1.14);x24 :find_root(h1(x)=0,x,-1.14,0);
x25 :find_root(h1(x)=0,x,0,1.1);x26 :find_root(h1(x)=0,x,1.2,2);x27 :find_root(h1(x)=0,x,2.4,18);
x28 :find_root(h1(x)=0,x,19,22);
(x22) - 2.678095068451025
(x23) - 1.5949142258885456
(x24) - 0.5356946897365676
(x25) 0.5454701726898044
(x26) 1.574684144021028
(x27) 2.6900577559822607
(x28) 19.84937567467901
```

10 Etude de h

```
(% i87) limit(h(x),x,minf);h(x22);h(x23);h(x24);h(x25);h(x26);h(x27);h(x28);limit(h(x),x,inf);
(%o79) min,f
(%o80) 95.77320731678024
(%o81) - 23.3518314737013
(%o82) 11.772041750207602
(%o83) - 14.071434917194342
(%o84) 18.268450423305236
(%o85) - 110.48663033305189
(%o86) 7.539452637700076108
(%o87) - ∞
```




FIGURE 7 – /fig7.png

Surprise : contrairement à ce que nous pensions précédemment, h s'annule 8 fois et non pas 7 fois, c'est à dire que les courbes se croisent en 8 points et non pas en 7 comme le graphique le suggérait. On s'aperçoit qu'il existe donc un huitième point mais dont la valeur est supérieure à 20, ce qui explique qu'il n'apparaissait pas sur le graphique.

h s'annule et change de signe 8 fois en x_{30} , x_{31} , x_{32} , x_{33} , x_{34} , x_{35} , x_{36} et x_{37} . $h(x)$ est négatif avant x_{30} (donc $f(x) < g(x)$ sur $]-\infty; x_{30}]$) puis change de signe après chaque racine, et donc la position des courbes de f et g s'inversent aussi à chaque fois. Trouvons ces valeurs :


```
(% i95) x30 :find_root(h(x)=0,x,-5,-2.7);x31 :find_root(h(x)=0,x,-2.7,-1.6);x32 :find_root(h(x)=0,x,-1.6,-0.5);
x33 :find_root(h(x)=0,x,-0.5,0.5);x34 :find_root(h(x)=0,x,0.5,1.5);x35 :find_root(h(x)=0,x,1.6,2.6);
x36 :find_root(h(x)=0,x,2.6,19);x37 :find_root(h(x)=0,x,20,30);
(x30) - 2.999930834791456
(x31) - 2.0011249027269153
(x32) - 0.992240478447905
(x33) - 0.027063080088140838
(x34) 1.058287348431664
(x35) 1.9370070271566375
(x36) 3.0268457522423007
(x37) 21.41929377684092
```

En conclusion les courbes f et g se coupent en 8 points et leurs positions respectives sont :

- sur $]-\infty, x_{30}]$ f dessous g
- sur $[x_{30}, x_{31}]$ f dessus g
- sur $[x_{31}, x_{32}]$ f dessous g
- sur $[x_{32}, x_{33}]$ f dessus g
- sur $[x_{33}, x_{34}]$ f dessous g
- sur $[x_{34}, x_{35}]$ f dessus g
- sur $[x_{35}, x_{36}]$ f dessous g
- sur $[x_{36}, x_{37}]$ f dessus g
- sur $[x_{37}, +\infty[$ f dessous g

11 Représentation de l'aire cherchée

```
(% i96) wxdraw2d(
title= ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[-5,7],
yrange=[-100,100],
filled_func=true,
fill_color=green,
filled_func=f(x),
explicit(g(x),x,x30,x36),
filled_func=false,
color=black,
line_width=1,
key= f(x) ,
explicit(f(x),x,-10,10),
color=blue,
key= g(x) ,
explicit(g(x),x,-10,10),
color=magenta,
point_type=7,
point_size=1,
points([[x30,f(x30)],[x31,f(x31)],[x32,f(x32)],[x33,f(x33)],[x34,f(x34)],[x35,f(x35)],[x36,f(x36)],[x37,f(x37)]]
)$
```



(%t96)

Etant donné l'échelle, on ne représente pas le dernier morceau d'aire car $x_{37}=21.4$ et $f(x_{37})=2 \times 10^9$ environ car le graphique deviendrait illisible. Comme nous connaissons la position des deux courbes, nous rajouterons cette valeur dans le calcul de l'aire.

12 Calcul de l'aire cherchée

Il y a 7 aires à calculer en fonction des positions respectives des courbes

```
(% i97) ratprint :false$
```

```
(% i98) A1 :float(integrate(f(x)-g(x),x,x30,x31));  
(A1) 57.206196329481514
```

```
(% i99) A2 :-float(integrate(f(x)-g(x),x,x31,x32));  
(A2) 14.85904535306223
```

```
(% i100) A3 :float(integrate(f(x)-g(x),x,x32,x33));  
(A3) 7.341161154688287
```

```
(% i101) A4 :-float(integrate(f(x)-g(x),x,x33,x34));  
(A4) 9.770068596698279
```

```
(% i102) A5 :float(integrate(f(x)-g(x),x,x34,x35));  
(A5) 10.260167577267717
```

```
(% i103) A6 :-float(integrate(f(x)-g(x),x,x35,x36));  
(A6) 70.48963425935528
```

```
(% i104) A7 :float(integrate(f(x)-g(x),x,x36,x37));  
(A7) 3.309453885094222109
```

A7 est l'aire non visible et non colorée sur le graphique.

```
(% i105) A :A1+A2+A3+A4+A5+A6+A7;  
(A) 3.3094540550204954109
```

On peut tenter d'intégrer directement avec la valeur absolue entre les valeurs limites de l'aire :

```
(% i107) integrate(abs(f(x)-g(x)),x,x30,x37);
```

```
(%o107)  $\int_{-2.999930834791456}^{21.41929377684092} |(x-3)(x-2)(x-1)x(x+1)(x+2)(x+3) - \%e^x| dx$ 
```

Maxima ne sachant pas calculer, on utilise la fonction de calcul numérique quad_qags

```
(% i108) quad_qags(abs(f(x)-g(x)),x,x30,x37);
```

```
(%o108) [3.3094540552746744109, 23.36183211459111, 273, 0]
```

On en conclut bien que l'aire cherchée vaut approximativement 3309454055 unités d'aire.